The author provides the ABAP code featured in his article for the program he wrote to streamline the procedure for reconciling flexible spending accounts (FSA). 

Specifications for the FSA reconciliation program

PURPOSE: Calculate and display aggregate plan year balance for FSA plans

SELECTION SCREEN: Optional - selection of one or more benefit plans. Required - plan year begin date.

INPUT: Read the last (as of run date) pay result for each employee. Process entries of table BENTAB from the selected results if BENTAB-PYBEG matches the selected plan year begin date and BENTAB-BPLAN is in the range of selected benefit plans.

PROCESSING: For selected BENTAB entries summarize BENTAB-EECMA and BENTAB-CLYTD by benefit plan.

OUTPUT: For each cumulated benefit plan display the sum of BENTAB-EECMA as ‘Contributions’, the sum of BENTAB-CLYTD as ‘Claims’ and the difference (BENTAB-EECMA – BENTAB-CLYTD) as ‘Fund Balance’.

ABAP code for the FSA reconciliation program

report zcrm_fsa_reconciliation line-count 65 line-size 80

       no standard page heading.

************************************************************************

*  Author:  Clay Molinari

*  Date:    12/14/2003

*  Purpose: Reconcile FSA funds after plan year end

************************************************************************

** This program calculates the difference between contributions made to

** and claims withdrawn from flexible spending plans.

** This program evaluates the surplus or

** shortfall of each participants account and produces a summary for the

** entire fund. It can be run at any time as a forecasting tool. But it

** should only be run after the grace period for prior year claims has

** expired if funds are to be withdrawn from the account.

************************************************************************

** Technical notes:

**   This program is intended for SAP version 4.5B and above. Earlier

**   versions will not contain table BENTAB in the payroll results

**   cluster. For earlier version of SAP the YTD contributions can be

**   read by wage type from the CRT table of the final pay result in

**   the prior year. The goal amount can be read from infotype 0170 and

**   the claims can be read from infotype 0172.

************************************************************************

tables: pa0001, pcl1, pcl2, pc261, pc261a, pa0170, t5uca.

include rpc2cd09.                 " cluster CD data definition

include rpc2ca00.                 " cluster CA data definition

include rpc2ruu0.                 " cluster RU data definition

include rpc2rx09.                 " cluster RU data definition

include rpppxd00.                 " data definition buffer pcl1/pcl2

include rpppxd10.                 " common part buffer pcl1/pcl2

include rpppxm00.                 " buffer handling routine

* records for report

data: begin of output_report occurs 0,

   barea        like pa0170-barea,

   bplan        like pa0170-bplan,

   contribs     like pc245-betrg,

   claims       like pc245-betrg,

end of output_report.

data:  true           type boolean value 'X',

       false          like true value ' ',

       plan_text like t5uca-ltext,

       balance        like pc245-betrg.

selection-screen begin of block p1 with frame title text-001.

select-options: p_bplan for pa0170-bplan.

parameters: p_datum like sy-datum obligatory.

selection-screen end of block p1.

initialization.

** default the date to the first day of the prior year. The

** program is most likely to be run 3 to 5 months after the

** plan year end.

  p_datum = sy-datum.

  p_datum(4) = p_datum(4) - 1.

  p_datum+4(4) = '0101'.

top-of-page.

    write: /(80) 'FSA RECONCILIATION' centered.

    write: 1 'Page: ', sy-pagno, 60 'Run Date: ', sy-datum.

    skip 1.

    write: /'Plan Year:', p_datum(4).

    write: /   'PLAN',

           32 'CONTRIBUTIONS',

           60 'CLAIMS',

           73 'BALANCE'.

  uline.

start-of-selection.

** select all employees as of the system date. Only

** those with FSA amounts will end up on the report. Even

** terminated employees are processed because they may have

** participated in the plan last year and terminated since then

  select * from pa0001

    where endda ge sy-datum and

          begda le sy-datum.

    perform get_pay_results.

  endselect.

end-of-selection.

  perform display_report.

form get_pay_results.

    clear rgdir. refresh rgdir.

    clear bentab. refresh bentab.

** clear the cluster buffer

    rp-init-buffer.

** set the CU cluster key

    cd-key-pernr = pa0001-pernr.

** Get list of pay results from cluster CU

    rp-imp-c2-cu.

** Position RGDIR at the most recent pay result. Table BENTAB

** contains entries for current and prior year accounts.

** using the most recent result assures that all claims have

** been recorded.

    loop at rgdir

         where void eq ' ' and

               reversal eq ' '.

    endloop.

    if sy-subrc eq 0.

** set the RU cluster key

      rx-key-pernr = pa0001-pernr.

      unpack rgdir-seqnr to rx-key-seqno.

** Get the pay result tables

      rp-imp-c2-ru.

** read bentab table for total contributions and claims

      loop at bentab

              where pybeg eq p_datum and

                    bplan in p_bplan.

        output_report-barea = bentab-barea.

        output_report-bplan = bentab-bplan.

        output_report-contribs = bentab-eecma.

        output_report-claims = bentab-clytd.

        collect output_report.

      endloop.

    endif.

endform.

form display_report.

  sort output_report.

  loop at output_report.

    skip 2.

    select ltext from t5uca into plan_text

      where langu = sy-langu

        and barea = output_report-barea

        and bplan = output_report-bplan.

    endselect.

    if sy-subrc ne 0.

       clear plan_text.

    endif.

    balance = output_report-contribs - output_report-claims.

    write: /   plan_text(20),

            25 output_report-contribs, '-',

            45 output_report-claims, '=',

            65 balance.

  endloop.

endform.                    " DISPLAY_REPORT

